eAppendix 1: Step-by-step implementation of TMLE
We describe the four steps of implementing TMLE in detail, although similar procedures can be also found in the literature.1,2
Step 1: Estimate an initial probability function of Y given A and W, denoted as . The standard logistic regression model is one possible approach:

Therefore, the initial probability can be estimated by: 
       (1)
Any terms that are functions of A and/or W can be included in the model, for example, polynomial terms of A and W as well as the interaction terms of A and W can be considered. Consequently, for each subject, the predicted probabilities for both counterfactual outcomes  and  can be attained as following by setting A=0 and A=1 for everyone respectively:


Step 2: Estimate a probability function of A given W, denoted as . A logistic regression model can again be used: 
  (2)
The probability of A given W is then estimated by:

A logistic model is used in step 1 and step 2 for demonstration. In fact, any suitable modeling approach can be employed such as other parametric models, semi- or non-parametric approaches including advanced machine learning techniques.3 
Step 3: Updating the density of Y given A and W.
This step aims to find a better prediction model targeted at minimizing mean squared error for the estimation of  and  by using the so called efficient influence function of  and .4 For  and , it has been shown that the parametric working model satisfied specific required conditions for the so-called fluctuation parameters  and .2 
The model is defined as following:

Where , and  are referred to as clever covariates, and  and  are estimated from step 2. The indicator function I(·) takes value one if its Boolean argument is true and value zero otherwise. The above parametric model is fitted with maximum likelihood estimation to obtain estimates for the fluctuation parameters (). This can be done with standard software by setting  as an offset in an intercept-free logistic regression with covariates  and .
Explicitly:      (3)
By substituting (), the estimated probability of Y given A and W and the probability of the counterfactual outcomes for each subject can be updated as follows:
so that,

This updating is performed by setting A=0 and A=1 for each subject in the probability functions , , as well as in the clever covariates  and . In general, TMLE is an iterative procedure, where  is replaced with , and  is updated until convergence (until  becomes sufficiently small), where the subscript  denotes the th iteration. However, for the parameters  and  in this case, convergence is achieved in one step. Therefore  is the only update needed. Model (3) targets  and  simultaneously by including both  and  in the model. Alternatively,  and  can be targeted separately by including the offset and  or and  in two different models. However, estimating  and  separately doubles this modeling step, and is thus less computationally favorable. 
It is also possible to target RD, RR, and OR directly by applying the corresponding efficient influence functions. With this approach, iterative updates may be needed to achieve the convergence of  when targeting RR and OR directly.
Step 4: Computing the substitution estimator of .
TMLE of  and  are given by the G-computation formula:5
 and .
In this case, the G-computation formula is implemented simply by averaging the estimated individual predicted probabilities from the respective updated outcome model .
The RD, RR, and OR (in the log scale) are then estimated by:





eAppendix 2: Standard errors, confidence intervals, and null-hypothesis testing in TMLE framework

TMLE for the population parameter are asymptotically normally distributed with mean  and variance , where  is the variance of the influence function (denoted as for : . TMLE is asymptotically linear under the assumptions .14 The corresponding Wald-type 95% confidence interval is given by: . The causal null hypothesis :  can be tested with the statistics: .6 
Efficient influence functions are parameter specific and take the following values for the three effect measure statistics of interest, where  denotes the last updated :





eAppendix 3: SAS routine for a general implementation of TMLE in binary point exposure and outcome study

/* Specify the analysis dataset in option: Dataset=
   Specify the name of the binary exposure variable in option: Var_exposure=
   Specify the name of the binary outcome variable in option: Var_outcome=
   List the covariates for the treatment model in option: Var_ps=
   List the treatment variable and the covariates for outcome model in the       
   option: Var_out= */

%macro estimation_TMLE(Dataset=,Var_exposure=,Var_outcome=,Var_ps=,Var_out=);

* Propensity score estimation – Treatment model in TMLE;
proc logistic data=&Dataset descending noprint; 
model &Var_exposure=&Var_ps;
output out=&Dataset(drop=_level_) pred=ps;
run;

* Compute clever covariates;
data &Dataset;
set &Dataset end=eof;
H1AW=&Var_exposure/ps;
H1W=1/ps;
H0AW=(1-&Var_exposure)/(1-ps);
H0W=1/(1-ps);
if eof=1 then call symput('n',_n_);
run;

* Outcome model in TMLE;
proc logistic data=&Dataset descending;
model &Var_outcome=&Var_out;
store Q_model;
output out=&Dataset xbeta=logit_Y;
run;

* Estimate fluctuation parameters;
ods output ParameterEstimates=epsilon;
proc logistic data=&Dataset descending;
model &Var_outcome=H1AW H0AW/offset=logit_Y noint;
quit;


* Compute the probability of the counterfactual outcomes;
data A_1;
set &Dataset;
&Var_exposure=1;
run;

data A_0;
set &Dataset;
&Var_exposure=0;
run;

proc plm source=Q_model noprint;
   score data=&Dataset out=&Dataset predicted=logit_Q_A;
run;

proc plm source=Q_model noprint;
   score data=A_0 out=A_0 predicted=logit_Q_A0;
run;

proc plm source=Q_model noprint;
   score data=A_1 out=A_1 predicted=logit_Q_A1;
run;

data &Dataset;
merge &Dataset A_1(keep=id logit_Q_A1) A_0(keep=id logit_Q_A0);
by id;
run;

data epsilon1;
set epsilon;
where variable ='H1AW';
run;

data epsilon0;
set epsilon;
where variable ='H0AW';
run;


* Update the probability of the counterfactual outcomes; 
* Estimate the parameter of interest, e.g. odds ratio;
* Estimate the standard error and 95% confidence interval;
proc iml;
use epsilon1; read all var{estimate} into epsilon1;
use epsilon0; read all var{estimate} into epsilon0;
use &Dataset; 
read all var{H1AW} into H1AW;
read all var{H0AW} into H0AW;
read all var{H1W} into H1W; 
read all var{H0W} into H0W;
read all var{&Var_exposure} into E;
read all var{&Var_outcome} into Y;
read all var{ps} into ps;
read all var{logit_Q_A} into logit_Q_A;
read all var{logit_Q_A0} into logit_Q_A0;
read all var{logit_Q_A1} into logit_Q_A1;

Q1_1W=epsilon1*H1W+logit_Q_A1;
Q1_0W=epsilon0*H0W+logit_Q_A0; 
Q1_AW=epsilon1*H1AW+logit_Q_A;

Y1=exp(Q1_1W)/(1+exp(Q1_1W));
Y0=exp(Q1_0W)/(1+exp(Q1_0W));
YA=exp(Q1_AW)/(1+exp(Q1_AW));
u1=mean(Y1);
u0=mean(Y0);
*log RD estimation;
effect="log RD";
est=u1-u0;
IC=(E/ps-(1-E)/(1-ps))#(Y-YA)+Y1-Y0-est;
Variance=var(IC)/&n;
create TMLE_EST var{effect est Variance};
append;
*log RR estimation;
edit TMLE_EST;
effect="log RR";
est=log(u1)-log(u0);
IC=(1/u1)#(E/ps#(Y-YA)+Y1-u1)-(1/u0)#((1-E)/(1-ps)#(Y-YA)+Y0-u0);
Variance=var(IC)/&n;
append;
*log OR estimation;
edit TMLE_EST;
effect="log OR";
est=log(u1/(1-u1))-log(u0/(1-u0)); 
IC=(1/u1+1/(1-u1))#(E/ps#(Y-YA)+Y1)-(1/u0+1/(1-u0))#((1-E)/(1-ps)#(Y-YA)+Y0);
Variance=var(IC)/&n;
append;
quit;

data TMLE_EST;
length var_ps $200.;
length var_out $200.;
var_ps="&var_ps";
var_out="&var_out";
set TMLE_EST;
Stderr=sqrt(Variance);
LowerCL=est-probit(0.975)*sqrt(Variance);
UpperCL=est+probit(0.975)*sqrt(Variance);
Method='TMLE';
rename est=Estimate;
run;
%mend;

eAppendix 4: Results of the sensitivity analysis with propensity score truncated at the 1st and 99th percentile


[bookmark: _GoBack]eTable 1 Effect estimate of statins use post-MI on 1-year all-cause mortality by different TMLE and IPW estimator modeling approaches with truncated propensity score at the 1st and 99th percentile
	Outcome Model
	A
	A, W1
	A, W2
	A, W3

	Method
	IPW1-4
	TMLE1-4
	TMLE5-8
	TMLE9-12
	TMLE13-16

	Treatment Model
	log(OR)
	SE(log(OR))
	OR
	log(OR)
	SE(log(OR))
	OR
	log(OR)
	SE(log(OR))
	OR
	log(OR)
	SE(log(OR))
	OR
	log(OR)
	SE(log(OR))
	OR

	NULL
	-1.12
	0.04
	0.30
	-1.2
	0.01
	0.31
	-0.85
	0.04
	0.43
	-0.57
	0.04
	0.57
	-0.32
	0.04
	0.73

	W1a
	-0.84
	0.04
	0.43
	-0.84
	0.04
	0.44
	-0.84
	0.04
	0.43
	-0.56
	0.04
	0.57
	-0.32
	0.04
	0.73

	W2b
	-0.47
	0.08
	0.62
	-0.49
	0.09
	0.61
	-0.51
	0.08
	0.60
	-0.58
	0.07
	0.56
	-0.36
	0.07
	0.70

	W3c
	-0.14
	0.09
	0.87
	-0.23
	0.11
	0.80
	-0.22
	0.01
	0.80
	-0.34
	0.09
	0.71
	-0.37
	0.08
	0.69


aW1: Predefined important confounders (age, sex, obesity, smoking, history of diabetes);
bW2: All pre-specified confounders;
cW3: All potential confounders (W2 + the 400 empirical selected variables).

